
1 [bookmark: _Toc132734390][bookmark: _Toc122434485]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk37 related to the title of this CR.
Contact:	Olivier Genoud
	olivier.genoud@etsi.org
2 [bookmark: _Toc122434488][bookmark: _Toc132734391][bookmark: _Toc295288959]Corrections required
2.1 [bookmark: _Toc106975634][bookmark: _Toc122623878][bookmark: _Toc109680267][bookmark: _Toc132734392][bookmark: _Toc325725665][bookmark: _Toc122434493][bookmark: _Toc295288970]Change 1
	Function name
	function f_UT_RequestPDUSession


	Reason for change
	If a timer is started immediately after sending an AT/MMI command, it may be started before the AT/MMI command has been sent

	Summary of change
	Add an optional parameter to wait for a local confirmation if the function is called before starting a timer.

Also implemented in:
f_UT_IMS_TriggerSMS
f_UT_RequestIMSEmergencyCall
f_UT_InitiateECall

	TTCN module
	UpperTesterFunctions.ttcn



  function f_UT_RequestPDUSession(UT_PTC_MTC_PORT p_Port,
                        integer p_SessionId,
                        charstring p_DNN := "", // @sic R5s100466 sic@
                        boolean p_EmergencyBearer := false,  // @sic R5-120403 sic@
                        integer p_SSCMode := 1,
                        charstring p_S_NSSAI_ForATCmd := "",
                        boolean p_Access3GPP := true,
                        template (omit) boolean p_RQos := omit,
                        template (omit) boolean p_MH6 := omit,
                        template (omit) boolean p_AlwaysOn := omit,
                        integer p_NSLPI := -1,
                        charstring p_SpecialType := "",
                        Ut_CnfReq_Type p_CnfRequired := CNF_REQUIRED)
  {
    var charstring v_Cmd := fl_BuildPDNRequest(p_DNN, p_EmergencyBearer);
    var charstring v_CommasToReqType := ",,,,,";
    var charstring v_CommasToNSLPI := ",,,";
    var charstring v_CommasToSSC := ",,,,,,";
    
    if (not p_EmergencyBearer) { // add commas to Request Type
      v_Cmd := v_Cmd & v_CommasToReqType;
    }
    // Then add commas to NSLPI
    v_Cmd := v_Cmd & v_CommasToNSLPI;
    if (p_NSLPI >= 0) {
      v_Cmd := v_Cmd & int2str(p_NSLPI);
    }
    // Then add commas to SSC
    v_Cmd := v_Cmd & v_CommasToSSC;
    v_Cmd := v_Cmd & int2str(p_SSCMode-1) & ",";
    v_Cmd := v_Cmd & """" & p_S_NSSAI_ForATCmd & """,";  //@sic R5s200501, R5s200963 sic@
    
    if (p_Access3GPP) {
      v_Cmd := v_Cmd & "0";
    } else {
      v_Cmd := v_Cmd & "1";
    }
    
    if (isvalue(p_RQos)) {
      if(valueof(p_RQos)) {
        v_Cmd := v_Cmd & ",1";
      } else {
        v_Cmd := v_Cmd & ",0";
      }
    }
    if (isvalue(p_MH6)) {
      if (not isvalue(p_RQos)) { // if previous field not included, need to add another comma
        v_Cmd := v_Cmd & ",";
      }
      if(valueof(p_MH6)) {
        v_Cmd := v_Cmd & ",1";
      } else {
        v_Cmd := v_Cmd & ",0";
      }
    }
    if (isvalue(p_AlwaysOn)) {
      if (not isvalue(p_RQos)) { // if previous field not included, need to add another comma
        v_Cmd := v_Cmd & ",";
      }
      if (not isvalue(p_MH6)) { // if previous field not included, need to add another comma
        v_Cmd := v_Cmd & ",";
      }
      if(valueof(p_AlwaysOn)) {
        v_Cmd := v_Cmd & ",1";
      } else {
        v_Cmd := v_Cmd & ",0";
      }
    }
    // Add comma if field is there or not
    v_Cmd := v_Cmd & ",";
    
    fl_RequestPDPContext(p_Port, p_SessionId, v_Cmd, p_SpecialType);  // @sic R5-227473 sic@
    fl_UT_ApplyCommand(p_Port, cas_UT_Req(ACTIVATE_BEARER, p_CnfRequiredNO_CNF_REQUIRED, cs_PDP_Cid(int2str(p_SessionId))));
  }
  
2.2 Change 2
	Function name
	functions fl_TC_10_1_3_2_Body (4 places)
fl_TC_10_1_8_1_Body (3 places)
fl_TC_10_1_8_2_Body (1 place)
fl_TC_10_1_8_4_Body (1 place)
fl_TC_10_1_8_5_Body (1 place)

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestPDUSession is followed by a timer starting

	TTCN module
	GSM_NR5GC.ttcn




	  function fl_TC_10_1_3_2_Body() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    var O1_Type v_PDUSessionIdToRelease;
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB2;
    var template (omit) DNN v_DNN;
    var integer v_InternetPduSessionID;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var charstring v_NSSAI := f_ConvertS_NSSAI_ForATCommand (cs_S_NSSAI_SST1eMBB);
    var template (value) NG_NAS_DL_Pdu_Type v_MsgToSend; // @sic R5-206355 sic@
    timer t_T3396 := 120.0; // @sic R5-201629 sic@
    timer t_T3584 := 120.0; // @sic R5-201629 sic@
    timer t_Wait := 5.0;

    v_InternetPduSessionID := f_NR5GC_MobileInfo_GetNoOfSessions()+1;

    //@siclog "Steps 1-2" siclog@
    //AT command to trigger a PDU session WITHOUT DNN (APN)
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, omit); // @sic R5s210148 sic@
    v_PDUSessionIdToRelease := f_NR_GetSessionId_ForLastPDUSession(); // @sic R5s210417 sic@

    //@siclog "Steps 3-4" siclog@
    f_NR5GC_PDUSessionRelease(v_NGC_NASCellA, // @sic R5s201131 sic@
                              v_PDUSessionIdToRelease,
                              cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                              cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00010'B)); //T3396 = 2 minutes // @sic R5-201629 sic@ // @sic R5-225052 sic@

    //@siclog "Steps 5" siclog@
    f_UT_RequestPDUSession(UT, v_InternetPduSessionID, -, -, -, v_NSSAI , -, -, -, -, -, -, LOCAL_CNF_REQUIRED); //AT command to trigger a PDU session WITHOUT DNN (APN)
    //@siclog "Steps 6" siclog@
    t_T3396.start; // @sic R5s200585 sic@
    alt{

        []SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered)))
            {
                t_T3396.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
            }
        []t_T3396.timeout
            {
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
            }
    }
    //@siclog "Steps 7-9" siclog@
    //AT command to trigger a PDU session WITHOUT DNN (APN)
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, omit); // @sic R5s200585, R5s201659, R5s210148 sic@
    v_PDUSessionIdToRelease := f_NR_GetSessionId_ForLastPDUSession(); // @sic R5s210417 sic@
    //@siclog "Steps 10-11" siclog@
    f_NR5GC_PDUSessionRelease(v_NGC_NASCellA, // @sic R5s201131 sic@
                              v_PDUSessionIdToRelease,
                              cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                              cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00000'B)); //T3396 = 0 minutes // @sic R5-201629 sic@

    //@siclog "Steps 12-14" siclog@
    //AT command to trigger a PDU session WITHOUT DNN (APN)
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, omit); // @sic R5s200585, R5s201659, R5s210148 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 13");
    //@siclog "Steps 15-16" siclog@
    v_PDUSessionIdToRelease := f_NR_GetSessionId_ForLastPDUSession(); // @sic R5s210585, R5s210417 sic@
    f_NR5GC_PDUSessionRelease(v_NGC_NASCellA, // @sic R5s201131 sic@
                              v_PDUSessionIdToRelease,
                              cs_GMM_GSM_Cause (omit, '00011010'B), //5GSM cause Insufficient resources
                              cs_GPRS_Timer2_3_IEI('37'O, '111'B, '00000'B)); //T3396 = Deactivated



    //@siclog "Steps 17" siclog@
    f_UT_RequestPDUSession(UT, v_InternetPduSessionID, -, -, -, v_NSSAI , -, -, -, -, -, -, LOCAL_CNF_REQUIRED); //AT command to trigger a PDU session WITHOUT DNN (APN)
    //@siclog "Steps 18" siclog@
    t_Wait.start;
    alt{

        []SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered)))
            {
                t_Wait.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 18");
            }
        []t_Wait.timeout
            {
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 18");
            }
    }
    //@siclog "Step 19" siclog@
    f_NR_UE_DeRegisterOnSwitchOff(v_NGC_NASCellA, STATE_CONNECTED_3A);
    //@siclog "Step 20" siclog@
    f_NR5GC_SwitchOnAndResetIMS(); //The UE is Switched ON @sic R5s211299 sic@
    //@siclog "Step 21" siclog@
    //Steps from Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed.
    f_NR5GC_RRC_IdleState1N_Def(v_NGC_NASCellA, -, noRrcConnectionRelease); // @sic R5s200585 sic@

    //@siclog "Steps 22-24" siclog@
    //AT command to trigger a PDU session WITHOUT DNN (APN)
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, omit); // @sic R5s200585, R5s201659, R5s210148 sic@
    v_InternetPduSessionID := v_InternetPduSessionID +1;
    //@siclog "Steps 25-26" siclog@
    //AT command to trigger a PDU session with S-NSSAI and DNN // @sic R5s201659 sic@
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, tsc_APN1_String, cs_S_NSSAI_SST1eMBB, ?); // @sic R5s200585, R5s201659, R5s210148 sic@
    v_PDUSessionIdToRelease := f_NR_GetSessionId_ForLastPDUSession(); // @sic R5s210417 sic@
    //Check that DNN is present in the PDU session established at Step 26
    v_DNN := f_Get_DNNForPDUSessionId(f_NR5GC_MobileInfo_GetSessionInfoList(), v_PDUSessionIdToRelease); //Store it for further use
    if (not isvalue(v_DNN)){
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "No DNN being sent in Step 26");
    }
    //@siclog "Steps 27-28" siclog@
    f_NR5GC_PDUSessionRelease(v_NGC_NASCellA, // @sic R5s201131 sic@
                              v_PDUSessionIdToRelease,
                              cs_GMM_GSM_Cause (omit, '01000011'B), //5GSM cause Insufficient resources for specific slice and DNN
                              cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00010'B)); //T3584 = 2 minutes // @sic R5-201629 sic@ // @sic R5-225052 sic@

    //@siclog "Steps 29" siclog@
    f_UT_RequestPDUSession(UT, v_InternetPduSessionID, tsc_APN1_String, -, -, v_NSSAI , -, -, -, -, -, -, LOCAL_CNF_REQUIRED); //AT command to trigger a PDU session with S-NSSAI and DNN // @sic R5s201659 sic@
    //@siclog "Steps 30" siclog@
    t_T3584.start; // @sic R5s200585 sic@
    alt{

        []SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered)))
            {
                t_T3584.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 30");
            }
        []t_T3584.timeout
            {
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 30");
            }
    }
    //@siclog "Steps 31-33" siclog@
    //AT command to trigger a PDU session with S-NSSAI and DNN // @sic R5s201659 sic@
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, tsc_APN1_String, cs_S_NSSAI_SST1eMBB, cs_S_NSSAI_SST1eMBB_WithIEI); // @sic R5s210148 sic@
    v_PDUSessionIdToRelease := f_NR_GetSessionId_ForLastPDUSession(); // @sic R5s210417 sic@
    //@siclog "Steps 34-35" siclog@
    f_NR5GC_PDUSessionRelease(v_NGC_NASCellA, // @sic R5s201131 sic@
                              v_PDUSessionIdToRelease,
                              cs_GMM_GSM_Cause (omit, '01000011'B), //5GSM cause Insufficient resources for specific slice and DNN
                              cs_GPRS_Timer2_3_IEI('37'O, '101'B, '00000'B)); //T3584 = 0 minutes



    //@siclog "Steps 36-38" siclog@
    //AT command to trigger a PDU session with S-NSSAI and DNN // @sic R5s201659 sic@
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, tsc_APN1_String, cs_S_NSSAI_SST1eMBB, cs_S_NSSAI_SST1eMBB_WithIEI); // @sic R5s200585, R5s201659, R5s210148 sic@
    v_PDUSessionIdToRelease := f_NR_GetSessionId_ForLastPDUSession(); // @sic R5s210417 sic@
    //@siclog "Steps 39-40" siclog@
    f_NR5GC_PDUSessionRelease(v_NGC_NASCellA, // @sic R5s201131 sic@
                              v_PDUSessionIdToRelease,
                              cs_GMM_GSM_Cause (omit, '01000011'B), //5GSM cause Insufficient resources for specific slice and DNN
                              cs_GPRS_Timer2_3_IEI('37'O, '111'B, '00000'B)); //T3584 = Deactivated

    //@siclog "Steps 41" siclog@
    f_UT_RequestPDUSession(UT, v_InternetPduSessionID, tsc_APN1_String, -, -, v_NSSAI , -, -, -, -, -, -, LOCAL_CNF_REQUIRED); //AT command to trigger a PDU session with S-NSSAI and DNN // @sic R5s201659 sic@
    //@siclog "Steps 42" siclog@
    t_Wait.start;
    alt{

        []SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered)))
            {
                t_Wait.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 42");
            }
        []t_Wait.timeout
            {
                f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 42");
            }
    }
    
    //@siclog "Steps 43" siclog@
    //Build the PDU Session Release
    v_MsgToSend := f_Get_NG_PDUSessionReleaseCommand(v_PDUSessionIdToRelease, cs_GMM_GSM_Cause (omit, '00100100'B), omit);
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, - , -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend))); // @sic R5-206355 sic@
    //@siclog "Steps 44" siclog@
    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp; // @sic R5-206355 sic@
    if (not f_Check_NG_GSMStatus (v_ReceivedAsp.Signalling.Nas[0].Pdu, v_PDUSessionIdToRelease, ?, cr_GMM_GSM_Cause(omit, tsc_NR5GCCause_InvalidPDUSessionIdentity))) {
       f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 44");
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 44");
    //@siclog "Step 45" siclog@
    f_NR_UE_DeRegisterOnSwitchOff(v_NGC_NASCellA, STATE_CONNECTED_3A);
    //@siclog "Step 46" siclog@
    f_NR5GC_SwitchOnAndResetIMS(); // @sic R5s211299 sic@
    //@siclog "Step 47" siclog@
    //Steps from Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed.
    f_NR5GC_RRC_IdleState1N_Def(v_NGC_NASCellA, -, noRrcConnectionRelease); // @sic R5s200585 sic@
    //@siclog "Steps 48-50" siclog@
    //AT command to trigger a PDU session with S-NSSAI and DNN // @sic R5s201659 sic@
    f_NR5GC_TriggerSinglePDUSession(v_NGC_NASCellA, v_InternetPduSessionID, -, tsc_APN1_String, cs_S_NSSAI_SST1eMBB, cs_S_NSSAI_SST1eMBB_WithIEI); // @sic R5s210148 sic@

    } //End of test body of TC_10_1_3_2
    



2.3 Change 3
	Function name
	Functions fl_TC_11_3_9_TestBody (1 place)
fl_TC_11_3_5_TestBody (2 places)

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestPDUSession is followed by a timer starting

	TTCN module
	UAC_NR5GC.ttcn




	  function fl_TC_11_3_9_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_RadioBearerId_Type v_SRB:= tsc_NR_RbId_SRB2;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (omit) DNN v_DNN;
    var boolean v_ReceiveRRCSetupRequest;
    var default v_MyDefaultVar:= null; // @sic R5-224717 sic@
    timer t_Wait := 11.2; // @sic R5-213507 sic@
    timer t_WaitTime := 5.0; // @sic R5-214783 sic@

    v_DNN := cs_DNN(-, f_DomainName_Encode(tsc_DNNValue)); // @sic R5s210307 sic@
    
    //@siclog "Step 3" siclog@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue, -, -, -, -, -, -, -, -, -, -, LOCAL_CNF_REQUIRED);
    
   //@siclog "Step 4" siclog@
   // Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is the same value in UL NAS TRANSPORT message in preamble within 20s?
    t_Wait.start;
    alt{
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
               // @sic R5s210307 sic@
               f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
            }
        []t_Wait.timeout
            {
               f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
            }
    }
    // @siclog "Step 5" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.9.3.2-1/2.
     v_CellPowerList_T1 := { //@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    
    if( f_GetTestcaseAttrib_NR_SNPN(testcasename())) //@sic R5-233351 sic@
    {
     //@siclog "Step 4-22a1" siclog@
     //Steps 2-20a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed
      f_NR5GC_MobileInfo_Init(); // Need to 'reset' all stored lists as all DRBs and PDU sessions do not exist in the new SNPN
      f_NR5GC_RRC_IdleState1N_Def (nr_Cell12);
    } else {
     // @siclog "Step 6" siclog@
     //The mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 12 and release RRC connection (Note 1)
     //Note 1:   SS ignore the PDU SESSION ESTABLISHMENT REQUEST sent before SS release RRC connection. @sic R5-224717 sic@
      v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(nr_Cell12));
      f_NR5GC_MobilityRegistration(nr_Cell12, -, -, -, -, noRrcConnectionRelease);
      f_NR_RRCRelease(nr_Cell12);
      deactivate(v_MyDefaultVar);
    }
 
    // @sic R5-214783 sic@
    // @siclog "Steps 6A - 6Bb1" siclog@
    t_WaitTime.start;
    alt{
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_MoData, ?)))
            {
              t_WaitTime.stop;
              f_NR_ULGrantConfiguration_Start(nr_Cell12, -, -, -, cs_NR_UplinkTimeAlignment_Stop, -, -, tsc_NoCnfReq ); // @sic R5s221054 sic@
              f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6Ba1");
            
            }
        [] t_WaitTime.timeout
            {
              //@siclog "Step 6Bb2" siclog@
              // Make the UE attempt to establish a PDU session for the DNN defined in Table 11.3.9.3.3-1. This can be done by an AT/MMI command.
                f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
                //@siclog "Step 6Bb3" siclog@
                //Check: Does the UE transmit a RRCSetupRequest message and establishmentCause is set to mo-Data?
                v_ReceiveRRCSetupRequest := f_NR_RRC_SetupRequest_Common( nr_Cell12, t_Wait, cr_NR_EstablishmentCause_MoData);
                if (v_ReceiveRRCSetupRequest) {
                    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6Bb3");
                } else {
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6Bb3");
                }
                
            }
    }
    
    //@siclog "Steps 9 - 14" siclog@
    //Steps 3-8 of the generic procedure for NR RRC_Connected specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                          tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0000'B)) {  // @sic R5s200654 R5s210307 sic@
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Steps 15 - 16" siclog@
    //The UE transmits a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is set to the same value with the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
    //The SS transmits PDU SESSION ESTABLISHMENT ACCEPT message.
    f_NR5GC_PDUSessionEstablishment (nr_Cell12, 1, -, -, - , v_DNN, -, v_DNN);
    
  }


2.4 Change 4
	Function name
	Functions fl_TC_9_1_10_3_TestBody (1 place)
fl_TC_9_1_10_4_TestBody (1 place)
fl_TC_9_1_10_6_TestBody (1 place)

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestPDUSession is followed by a timer starting

	TTCN module
	NetworkSlicingEnh_NR5GC.ttcn




	  function fl_TC_9_1_10_3_TestBody() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_CellsOnDifferentPLMN);
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NG_NAS_UL_Message_Type v_GSMMsg;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_s;
    var ProcedureTransactionIdentifier v_PTI := '00'O;
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1;       //@sic R5s220948 sic@
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (v_NGC_NASCellA);
    var template (value) NG_MobileIdentity v_GUTIToSend := f_NG_GutiParameters2MobileIdentity (f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA)), v_GutiParams);
    var TrackingAreaCode v_TAC := f_NR_CellInfo_GetTAC (v_NGC_NASCellA);
    var template (value) NG_TrackingAreaIdList v_TAIList := cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA)), {bit2oct(v_TAC)});
    var template (value) RejectedNSSAI v_RejectedNSSAI := {
      iei := '11'O,
      iel := '02'O,        //@sic R5-224165 sic@
      rejectS_NSSAI := {cs_RejectedS_NSSAI('0001'B, '0010'B, '01'O, omit)}};    //@sic R5-222008 R5-223429 R5-224165 sic@
    var template (value) NSSAI v_AllowedNSSAI:= { // @sic R5s221179 Baseline Moving sic@
      iei := '15'O,
      lvPart := {
        iel := '06'O,
        s_nssai := {cs_S_NSSAI('01'O, '02'O, omit), cs_S_NSSAI('01'O, '03'O, omit), cs_S_NSSAI('01'O, '04'O, omit)}     //@sic R5-223429 sic@
      }};
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var template (value) NG_NAS_DL_Pdu_Type v_MToSend;
    var charstring v_AT_Response;
    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA));
    var charstring v_NSSAI := f_ConvertS_NSSAI_ForATCommand (cs_S_NSSAI('01'O, '01'O, omit));
    var template (value) NSSAI v_ConfiguredNSSAI := { // @sic R5s221179 Baseline Moving sic@
      iei := '31'O,
      lvPart := {
        iel := '02'O,
        s_nssai := {cs_S_NSSAI('01'O, '01'O, omit)}
      }};
    var template (value) NSSAI v_PendingNSSAI := { // @sic R5s221179 Baseline Moving sic@
      iei := '39'O,
      lvPart := {
        iel := '02'O,
        s_nssai := {cs_S_NSSAI('01'O, '01'O, omit)}
      }};
    var template (value) GPRS_Timer3 v_T3512 := cs_GPRS_Timer2_3_IEI('5E'O, '101'B, '00001'B);

    //@siclog "Step 0A" siclog@
    f_NR5GC_DeleteAllStoredNSSAI(v_NGC_NASCellA);   // @sic R5-227438 sic@
    
    //@siclog "Step 0B" siclog@
    f_NR_UE_DeRegisterOnSwitchOff (v_NGC_NASCellA, STATE_CONNECTED_3A);  // @sic R5-227438 sic@

    //@siclog "Step 1" siclog@
    f_NR5GC_SwitchOnAndResetIMS();
    
    //@siclog "Step 2" siclog@
    //The UE transmits a REGISTRATION REQUEST message including NSSAA bit
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure,-,-,-,-,cdr_NG_GMM_Cap_NSSAA)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }else {
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    }
    
    //@siclog "Step 3-11" siclog@
    f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo,
                                v_NGC_NASCellA, v_ReceivedMsg,
                                f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA)),
                                TESTMode_OFF);
        
    //@siclog "Step 12" siclog@
    //The SS transmits a REGISTRATION ACCEPT message including Rejected S-NSSAI=1
    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,
                                                   v_GUTIToSend,
                                                   v_TAIList,
                                                   omit,
                                                   v_AllowedNSSAI,  //@sic R5-222008 sic@
                                                   v_RejectedNSSAI, -, -, -,
                                                   -, -, -, -, -, -, -, -, -, -,
                                                   -, -, -, -, -, -, -, -, -, -,
                                                   -, '1'B);
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    
    //@siclog "Step 13" siclog@
    f_NR5GC_Registration_Complete(v_NGC_NASCellA);
    
    //@siclog "Step 14a" siclog@
    if (pc_noOf_PDUsSameConnection > 0 ){
      f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, -, tsc_NR_RbId_SRB1);
      v_SRB := tsc_NR_RbId_SRB2;   //@sic R5s220948 sic@
    }

    //@siclog "Step 14A" siclog@
    f_NR_RRCRelease(v_NGC_NASCellA); //@sic R5-232339 sic@
    
    //@siclog "Step 15" siclog@
    //Check: Is S-NSSAI=1 in the Rejected NSSAI list with cause "S-NSSAI not available due to the failed or revoked network slice-specific authentication and authorization" associated with current PLMN using AT/MMI command (+C5GNSSAIRDP)?
    v_AT_Response := f_UT_ReadRejectedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN));
    if (match (v_AT_Response, f_Convert_ListOfRejectedNSSAI_ForATCommand(v_RejectedNSSAI))) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 15");
    }

    //@siclog "Step 16" siclog@
    //Cause the UE to request establishment of PDU session with S-NSSAI=1
    f_UT_RequestPDUSession(UT, 1, -, -, -, v_NSSAI , -, -, -, -, -, -, -, LOCAL_CNF_REQUIRED);
    
    //@siclog "Step 17" siclog@
    //Check: Does the UE transmit a  RRCSetupRequest message within 5 seconds?
    f_NR_CheckNoRRCSetupReq(v_NGC_NASCellA, 5.0);   //@sic R5-232339 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 17");
    
    //@siclog "Step 18" siclog@
    //Switch off procedure in RRC_IDLE specified in TS 38.508-1 subclause 4.9.6.1 is performed.
    f_NR_UE_DeRegisterOnSwitchOff(v_NGC_NASCellA,
                                  STATE_IDLE_1A,    //@sic R5-232339 sic@
                                  v_SRB);  //@sic R5s220948 sic@
    
    //@siclog "Step 19" siclog@
    f_NR_SetCellPower (v_NGC_NASCellA, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);      //@sic R5-222008 sic@
    
...


2.5 Change 5
	Function name
	Functions fl_TC_9_1_12_1_TestBody (2 places)
fl_TC_9_1_12_2_TestBody (1 place)
fl_TC_9_1_12_3_TestBody (1 place)
fl_TC_9_1_12_4_TestBody (1 place)
fl_TC_9_1_12_5_TestBody (1 place)

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestPDUSession is followed by a timer starting

	TTCN module
	NSAC_NR5GC.ttcn




	  function fl_TC_9_1_12_1_TestBody() runs on NR5GC_PTC
  {
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_CellsOnDifferentPLMN);
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (v_NGC_NASCellA);
    var template (value) NG_MobileIdentity v_GUTIToSend := f_NG_GutiParameters2MobileIdentity (f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA)), v_GutiParams);
    var TrackingAreaCode v_TAC := f_NR_CellInfo_GetTAC (v_NGC_NASCellA);
    var template (value) NG_TrackingAreaIdList v_TAIList := cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA)), {bit2oct(v_TAC)});
    var template (value) NSSAI v_AllowedNSSAI_Step12:= {
      iei := '15'O,
      lvPart := {
        iel := '02'O,
        s_nssai := {cs_S_NSSAI('01'O, '03'O, omit)}
      }};
    var template (value) NSSAI v_ConfiguredNSSAI := {
      iei := '31'O,
      lvPart := {
        iel := '04'O,
        s_nssai := {cs_S_NSSAI('01'O, '01'O, omit), cs_S_NSSAI('01'O, '02'O, omit)}
      }};
    var template (value) PartialXtdRejectedNSSAI v_PartialXtdRejectedNSSAI1 := f_BuildNR5GC_PartialXtdRejectedNSSAIType1({cs_XtdRejectedS_NSSAI('0001'B, '0011'B, '01'O, omit, omit, omit)}, '82'O);   //@sic R5-233378 sic@
    var template (value) PartialXtdRejectedNSSAI v_PartialXtdRejectedNSSAI2 := f_BuildNR5GC_PartialXtdRejectedNSSAIType1({cs_XtdRejectedS_NSSAI('0001'B, '0011'B, '02'O, omit, omit, omit)}, '21'O);   //@sic R5-233378 sic@
    var template (value) ExtdRejectedNSSAI v_ExtdRejectedNSSAI := f_BuildNR5GC_ExtdRejectedNSSAI({v_PartialXtdRejectedNSSAI1, v_PartialXtdRejectedNSSAI2});
    var NR_RadioBearerId_Type v_SRB := tsc_NR_RbId_SRB1;
    var charstring v_AT_Response;
    var charstring v_NSSAI := f_ConvertS_NSSAI_ForATCommand (cs_S_NSSAI('01'O, '01'O, omit));
    var charstring v_NSSAI2 := f_ConvertS_NSSAI_ForATCommand (cs_S_NSSAI('01'O, '02'O, omit));
    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA));
    timer t_T3526 := 720.0;

    //@siclog "Step 1" siclog@
    f_NR5GC_SwitchOnAndResetIMS();
    
    //@siclog "Step 2" siclog@
    //Check: Does UE transmit a REGISTRATION REQUEST message including ER-NSSAI bit?
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_Secure,-,-,-,-,cdr_NG_GMM_Cap_ERNSSAI)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }else {
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
    }
    
    //@siclog "Step 3-11" siclog@
    f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo,
                                v_NGC_NASCellA, v_ReceivedMsg,
                                f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellA)),
                                TESTMode_OFF);
        
    //@siclog "Step 12" siclog@
    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,
                                                   v_GUTIToSend,
                                                   v_TAIList,
                                                   omit,
                                                   v_AllowedNSSAI_Step12,
                                                   -, v_ConfiguredNSSAI, -, -,
                                                   -, -, -, -, -, -, omit, -, -, -,
                                                   -, -, -, -, -, -, -, -, -, -,
                                                   -, -, -, -, -, -, -, -, -, -,
                                                   -, -, -, -, v_ExtdRejectedNSSAI);
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA, tsc_NR_RbId_SRB1, -, cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    t_T3526.start;
    
    //@siclog "Step 13" siclog@
    f_NR5GC_Registration_Complete(v_NGC_NASCellA);
    
    //@siclog "Step 14a" siclog@
    if (pc_noOf_PDUsSameConnection > 0 ){
      f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, -, v_SRB);
      v_SRB := tsc_NR_RbId_SRB2; // SRB2 is established during PDU Session Establishment, hence further NAS signalling will happen on SRB2
    }
 
    //@siclog "Step 15" siclog@
    f_NR_RRCRelease(v_NGC_NASCellA);
    
    //@siclog "Step 16" siclog@
    //Check: Are S-NSSAI=1 and S-NSSAI=2 in the Rejected NSSAI list with cause 'S-NSSAI not available due to maximum number of UEs reached' associated with current PLMN using AT/MMI command (+C5GNSSAIRDP)?
    v_AT_Response := f_UT_ReadXtdRejectedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN)); // @sic R5s230702 sic@
    if (f_MatchXtdRejectedNSSAI_WithATCommand({cs_XtdRejectedS_NSSAI('0001'B, '0011'B, '01'O, omit, omit, omit),
                                          cs_XtdRejectedS_NSSAI('0001'B, '0011'B, '02'O, omit, omit, omit)},
                                          v_AT_Response)) {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");
    } else {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 16");
    }
    
    //@siclog "Step 17" siclog@
    //Cause the UE to request establishment of PDU session with S-NSSAI=1
    f_UT_RequestPDUSession(UT, 1, -, -, -, v_NSSAI , -, -, -, -, -, -, -, LOCAL_CNF_REQUIRED);
    
    //@siclog "Step 18" siclog@
    //Check: Does the UE transmit an RRCSetupRequest message within T3526 associated with S-NSSAI=1 in the next 15 seconds?
    f_NR_CheckNoRRCSetupReq(v_NGC_NASCellA, 15.0);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 18");

    //@siclog "Step 19" siclog@
    //Cause the UE to request establishment of PDU session with S-NSSAI=2
    f_UT_RequestPDUSession(UT, 1, -, -, -, v_NSSAI2 , -, -, -, -, -, -, -, LOCAL_CNF_REQUIRED);
    
    //@siclog "Step 20" siclog@
    //Check: Does the UE transmit an RRCSetupRequest message within T3526 associated with S-NSSAI=2 in the next 15 seconds?
    f_NR_CheckNoRRCSetupReq(v_NGC_NASCellA, 15.0);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 20");

    //@siclog "Step 21" siclog@
    t_T3526.timeout;

    //@siclog "Step 22" siclog@
    //Check: Using AT/MMI command (+C5GNSSAIRDP) verify that S-NSSAI=1 is not in the Rejected NSSAI list associated with current PLMN
    v_AT_Response := f_UT_ReadXtdRejectedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN)); // @sic R5s230702 sic@
    if (f_MatchXtdRejectedNSSAI_WithATCommand({cs_XtdRejectedS_NSSAI('0001'B, '0011'B, '01'O, omit, omit, omit)},
                                          v_AT_Response)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 22")
    } else {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 22");
    }

    //@sic R5-233378 sic@
    
    //@siclog "Step 23" siclog@
    f_NR_UE_DeRegisterOnSwitchOff(v_NGC_NASCellA, STATE_IDLE_1A);
    
    //@siclog "Step 24" siclog@
    f_NR_SetCellPower (v_NGC_NASCellA, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE);     //@sic R5-231553 sic@
    
    //@siclog "Step 25" siclog@
    f_NR5GC_SwitchOnAndResetIMS();
    
    //@siclog "Step 26" siclog@
    v_AT_Response := f_UT_ReadXtdRejectedNSSAI(UT, f_ConvertPLMNtoString(v_PLMN)); // @sic R5s230702 sic@
    if (f_MatchXtdRejectedNSSAI_WithATCommand({cs_XtdRejectedS_NSSAI('0001'B, '0011'B, '02'O, omit, omit, omit)},
                                          v_AT_Response)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 26")
    } else {
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 26");
    }
    
    //@sic R5-233378 sic@
    //delete step 27-29
    }//End of fl_TC_9_1_12_1_Body



2.6 Change 6
	Function name
	Function fl_TC_11_3_1_TestBody

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_IMS_TriggerSMS is followed by a timer starting

	TTCN module
	UAC_NR5GC.ttcn




	  function fl_TC_11_3_1_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var float v_TimerValue := 21.0; //@sic R5-221459 sic@
    timer t_WaitTime;
    
    
    //@sic R5s211658 sic@
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1, //@sic R5-221459 sic@
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(4,1),
                                         -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '0000000'B)));
    f_NR_ModifySysinfo(nr_Cell1);
    f_NR_WaitModificationPeriods(nr_Cell1);

    
    // @sic R5-216192 sic@
    //@siclog "Step 13" siclog@
    //Make the UE attempt a MMTel voice call.
    f_UT_RequestIMSCall(UT);
     
    //@siclog "Step 14" siclog@
    //Check: Does the UE transmit RRCSetupRequest message including mo-VoiceCall with within 21 s?
    t_WaitTime.start(v_TimerValue);
     alt {
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(mo_VoiceCall,?)))
        {
          t_WaitTime.stop;
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
        }
      [] t_WaitTime.timeout
        {
        }
    }
   
    //@siclog "Step 15" siclog@
    //SS changes SIB1 according to Table 11.3.1.3.3-1 and sends Short Message on PDCCH using P-RNTI. Wait for the new system information to take effect and T390 expire. (Note 1)
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(6,1),
                                         -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '0000000'B)));
                                        
    f_NR_ModifySysinfo(nr_Cell1); //Modify Sysinfo in SS
    
    //@sic R5-221459 sic@
    //@siclog "Step 15A" siclog@
    t_WaitTime.start(20.0);
     alt {
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(mo_VoiceCall,?)))
        {
          t_WaitTime.stop;
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15Aa1");
          f_NR_ULGrantConfiguration_Start(nr_Cell1); // @sic R5s220817 sic@
        }
      [] t_WaitTime.timeout
        {
          //@siclog "Step 15Ab2 - 15Ab3" siclog@
          //Make the UE attempt another MO MTSI MMTEL Voice session.
          //Check: Does the UE send NR RRCSetupRequest with EstablishmentCause set to 'mo-VoiceCall'?
          f_UT_RequestIMSCall(UT);
          f_NR_RRC_SetupRequest_Def(nr_Cell1, mo_VoiceCall);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15Ab3");
        }
    }

     
    //@siclog "Steps 16-17" siclog@
    // void
    
    //@siclog "Steps 18-28" siclog@
    //Step 3 to step 13 from test procedure for IMS MO speech call establishment as described in TS 38.508-1 [4] Table 4.9.15.2.2-1 take place,

    //Send RRC Connection Setup on SRB0 (CCCH): done automatically by SS in Msg4 of the RA procedure
    
    // receive RRC Connection Setup Complete with piggy-backed NAS message on SRB1 (DCCH):
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                          tsc_SHT_IntegrityProtected
                                          );
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)), '0001'B)) {  // @sic R5s200654 sic@
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
    
    v_GSM_MobilityInfo:= f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), IMS_DNN);
    
    // done on IMS PTC
    // Wait until need to add voice QoS
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1], "Add Voice Bearer"); // @sic R5-225362 sic@
    f_NR5GC_ModifyPDUSession_AddVoice(nr_Cell1, valueof(v_GSM_MobilityInfo));
    if (not pc_Preconditions) {  // @sic R5s220579 sic@
      f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IMS_IPCAN_Trigger("Voice Bearer Added"));  // @sic R5-225362 sic@
    }
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Voice Call Established");

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1], "Voice Call Established"); // @sic R5-225362 sic@
    
    //@siclog "Steps 29-31" siclog@
    //Step 1 to step 5 from test procedure for test procedure for IMS MO speech call release as described in TS 38.508-1 [4] Table 4.9.17.2.2-1 take place
    f_NR5GC_CallRelease(nr_Cell1, IPCAN_SpeechCall);
    
    //@siclog "Step 31A" siclog@
    //The SS transmits an RRCRelease message to release RRC connection and move the UE to RRC_IDLE.
    f_NR_RRCRelease(nr_Cell1); //@sic R5-216192 sic@
  
    //@siclog "Step 32" siclog@
    //AT command to make the UE attempt to send SMS over IP
    f_UT_IMS_TriggerSMS(UT, LOCAL_CNF_REQUIRED);
    
    //@siclog "Step 33" siclog@
    //Check: Does the UE transmit an RRCSetupRequest message with 'establishmentCause' set to 'mo-SMS'?
    // @sic R5-210356 sic@
    t_WaitTime.start(v_TimerValue);
    alt {
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mo_SMS))) //@sic R5-221459 sic@
      {
         t_WaitTime.stop;
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 33");
      }
      [] t_WaitTime.timeout {}
    }
    
    //@siclog "Step 34" siclog@
    //SS changes SIB1 according to Table 11.3.1.3.3-1 and sends Short Message on PDCCH using P-RNTI. Wait for the new system information to take effect and T390 expire. (Note 1)
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1, omit);
    f_NR_ModifySysinfo(nr_Cell1);
   
   //@sic R5-221459 sic@
    //@siclog "Step 34A" siclog@
    t_WaitTime.start(20.0);
     alt {
     //@siclog "Step 34Aa1" siclog@
     //Check: Does the UE send NR RRCSetupRequest with EstablishmentCause set to 'mo-SMS'?
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest( mo_SMS,?)))
        {
          t_WaitTime.stop;
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 34Aa1");
          f_NR_ULGrantConfiguration_Start(nr_Cell1); // @sic R5s220817 sic@
        }
      [] t_WaitTime.timeout
        {
           //@siclog "Step 34Ab2" siclog@
          //AT command to make the UE attempt to send another SMS over IP
          f_UT_IMS_TriggerSMS(UT);
          f_NR_RRC_SetupRequest_Def(nr_Cell1, mo_SMS);
          f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 34Ab3");
        }
    }
    
                                       
    //@siclog "Steps 36-43" siclog@
    //Steps 3-16 from generic procedure IMS MO SMS in 5GS as described in TS 38.508-1 [4] Table 4.9.19.2.2-1 take place.

    // Send RRC Connection Setup on SRB0 (CCCH): done automatically by SS in Msg4 of the RA procedure
    
    // receive RRC Connection Setup Complete with piggy-backed NAS message on SRB1 (DCCH):
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                          tsc_SHT_IntegrityProtected
                                          );
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)), '0001'B)) {  // @sic R5s200654 sic@
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); //@sic R5s211658 sic@
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
    
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
  
  }



2.7 Change 7
	Function name
	Functions fl_TC_11_3_2_TestBody (1 place)
fl_TC_11_3_7_TestBody (1 place)

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestIMSEmergencyCall is followed by a timer starting

	TTCN module
	UAC_NR5GC.ttcn




	  function fl_TC_11_3_2_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var RejectWaitTime v_RejectWaitTime := 16;
    var float v_WaitForConnectionMin := f_NR_SetTimerToleranceMin(nr_Cell1, rrcTimer, int2float(v_RejectWaitTime));
    var float v_TimerValue := 30.0;
    timer t_WaitTime;
    timer t_Wait:= 15.0;

    //@siclog "Step 1" siclog@
    // Make the UE attempt an emergency call dialling a number which is stored on the ME (e.g. 112 or 911). (Note 1)
    f_UT_RequestIMSEmergencyCall (UT , -, LOCAL_CNF_REQUIRED);

    //@siclog "Step 2" siclog@
    //Check: Does the UE transmit an RRCSetupRequest message with establishmentCause set to 'emergency' within 30 s'?
    f_NR_CheckNoRRCSetupReq(nr_Cell1, v_TimerValue, cr_NR_EstablishmentCause_emergency, "Step 2");
    
    //@sic R5-213505 sic@
    //@siclog "Step 2A1" siclog@
    //IF pc_SwitchOnOff THEN switch off UE, ELSE IF pc_USIM_Removal THEN remove the USIM, ELSE power off UE in which case steps 2A2 to 2A4 are not performed
    
     if(pc_SwitchOnOff){
      f_UT_SwitchOffUE(UT, true);

      //@siclog "Step 2A2" siclog@
      //SS starts timer1 = 5 sec to see if UE performs optional Step 10a3a1 (Note 5)
      t_WaitTime.start(15.0);//@sic R5-216191 sic@

      //@siclog "Steps 2A3a1-2A3b1" siclog@
      //@sic R5-217001 sic@
        alt{
      
            [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest())) -> value v_ReceivedAsp
            {
                t_WaitTime.stop;
                f_NR_RRCSetup_Def (nr_Cell1);
                v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                                cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                                tsc_SHT_IntegrityProtected);
                f_NR5GC_OptionalIMSDeregisterOnSwitchOff(nr_Cell1, v_ReceivedMsg);
                f_NR_RRCRelease(nr_Cell1);
            }
            [] t_WaitTime.timeout{ //@sic R5-216191 sic@
            }
        }
    }

    //@siclog "Step 2A5" siclog@
    //ELSE power off UE
    else {
      f_UT_PowerOffUE (UT);
      

    }
    
    f_NR5GC_MobileInfo_Init(); //@sic R5s210757 sic@
    
    //@siclog "Step 2A5" siclog@
    //Switch on the UE
    f_NR5GC_SwitchOnAndResetIMS(); //@sic R5s210757 sic@
    
    //@siclog "Step 2A6" siclog@
    //The UE performs Registration procedure as specified in TS 38.508-1 [4] subclause 4.5.2.
    f_NR5GC_RRC_IdleState1N_Def(nr_Cell1);
    
    //@siclog "Step 3" siclog@
    //SS changes SIB1 according to Table 11.3.2.3.3-2 and sends Short Message on PDCCH using P-RNTI. Wait for 2.1* modification period to allow the new system information to take effect and T390 expire. (Note 1)
    // Table 11.3.2.3.3-2: SIB1 for NR Cell 1
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_7Entry, //@sic R5-205321 sic@
                                         -, cs_UAC_BarringInfoSetList_1Entry(p00, s4, '0000000'B)));
                                        
    f_NR_ModifySysinfo(nr_Cell1); //Modify Sysinfo in SS
    f_NR_WaitModificationPeriods(nr_Cell1);
     
    //@siclog "Step 4" siclog@
    //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);
    
    //@siclog "Step 5" siclog@
    //Check: Does the UE transmit an RRCSetupRequest message with establishmentCause set to 'mt-access' ?
    f_NR_RRC_SetupRequest_Def(nr_Cell1, mt_Access);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    
    //@siclog "Steps 6-12" siclog@
    // Steps 3 to 9a1 of the generic test procedure in TS 38.508-1 [4] Table 4.9.4.2.2-1 are performed on NR Cell 1.
    f_NR_RRCSetup_Def (nr_Cell1);
    
    // receive RRC Connection Setup Complete with piggy-backed NAS message on SRB1 (DCCH):
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                          tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0010'B)) {  // @sic R5s200654 sic@
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell1, v_ReceivedMsg);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Check 5GC RRC_IDLE");
  
    f_NR_RRCRelease(nr_Cell1);
    //@siclog "Step 13" siclog@
    // SS changes SIB1 according to 38.508-1 [4] Table 4.6.1-28 and sends Short Message on PDCCH using P-RNTI. Wait for 2.1* modification period to allow the new system information to take effect.
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1, omit);
    f_NR_ModifySysinfo(nr_Cell1);
    f_NR_WaitModificationPeriods(nr_Cell1);

    //@siclog "Steps 14-15" siclog@
    //Steps 1-2 of the NR RRC_CONNECTED procedure in table 4.5.4.2-3 in TS 38.508-1[4] are performed.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);
    f_NR_RRC_SetupRequest_Def(nr_Cell1);
    
    //@siclog "Step 16" siclog@
    //The SS responds with RRCReject message with IE waitTime set to 10s.
    SRB.send(cas_NR_SRB0_RrcPdu_REQ(nr_Cell1,
                                    -,
                                    cs_38508_RRCReject(v_RejectWaitTime)));
                                    
    t_WaitTime.start(v_WaitForConnectionMin);
                                    
    //@siclog "Step 17" siclog@
    // Make the UE attempt an emergency call dialling a number which is stored on the ME (e.g. 112 or 911). (Note 1)
    //NOTE 1: This could be done by e.g. MMI or AT command.
    f_UT_RequestIMSEmergencyCall (UT);
    
    t_WaitTime.stop; //@sic R5s210804 sic@
    
    //@sic R5-216191 sic@
    //EXCEPTION : Steps 21a1 - 21b6 describe the UE behaviour that depends on UE capability; the "lower case letter" identifies a step sequence that takes place if a capability is supported
    //@siclog "Steps 21a1 - 21a12" siclog@
    //IF pc_NR_5GC_EmergencyServices  THEN Step 2-13 for IMS Emergency call establishment are performed as specified in TS 38.508-1 [4], subclause 4.9.11 on NR Cell 1.
    if (pc_NR_5GC_EmergencyServices){  //@sic R5-221456 sic@
        f_NR5GC_EmergencyCallWithRegistration(nr_Cell1);
        //@siclog "Step 21a13 - 21a14" siclog@
        //Make the UE release the emergency call.
        //Test procedure for IMS MO Emergency call release is performed as specified in TS 38.508-1 [4], subclause 4.9.12A on NR Cell 1.
        f_NR5GC_EmergencyCallRelease(nr_Cell1);
        f_NR_RRCRelease(nr_Cell1);
        f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(PDN_2));
    }
    else{
        //@siclog "Steps 21b1- 21b4" siclog@
        //ELSE Step 2- 5 for IMS Emergency call establishment are performed as specified in TS 38.508-1 [4], subclause 4.9.11 on NR Cell 1.
        f_NR_RRC_SetupRequest_Common(nr_Cell1, t_Wait, cr_NR_EstablishmentCause_emergency);
        f_NR_RRCSetup_Def(nr_Cell1);
        v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell1,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);


        f_NR_RRC_ActivateSecurity (nr_Cell1, valueof(v_ReceivedMsg.SecurityProtection.NasCount));
        //@siclog "Steps 21b5" siclog@
        //IF pc_SwitchOnOff THEN switch off UE, IF pc_USIM_Removal THEN remove the USIM.
        if(pc_SwitchOnOff){
          f_UT_SwitchOffUE(UT, true); //@sic R5s220418 sic@
            //@siclog "Steps 21b6" siclog@
            t_WaitTime.start(15.0);
            alt{
                //@siclog "Steps 21b6a1" siclog@
                [] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1,
                                             tsc_NR_RbId_SRB1,
                                             cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp{
                                             
                    t_WaitTime.stop;
                    if (not f_Check_NG_DeregistrationReqMOMsg (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, crs_DeregisterType('1'B, '0'B, '01'B))) {
                        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Deregistration Request Message Failed");
                    }
                    //@siclog "Steps 21b6a2" siclog@
                    f_NR_RRCRelease(nr_Cell1);
                }
                [] t_WaitTime.timeout {
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Deregistration failed");
         
                }
            }
        
        }
        else{
            //@siclog "Steps 21b6b1" siclog@
            f_UT_PowerOffUE (UT);
        }
        
        
    }
  }


2.8 Change 8
	Function name
	Function fl_TC_11_5_5_TestBody 


	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_InitiateECall is followed by a timer starting

	TTCN module
	ECall_NR5GC.ttcn




	  function fl_TC_11_5_5_TestBody() runs on NR5GC_PTC
  {
    var float v_Wait:= 120.0;
    
    //@siclog "Step 1" siclog@
    f_NR5GC_SwitchOnAndResetIMS();
    
    //@siclog "Step 2" siclog@
    f_Delay(60.0);
    f_NR5GC_RRC_IdleState1N_Def(nr_Cell1);   // @sic R5-227244 sic@
    
    //@siclog "Step 3" siclog@
    f_UT_InitiateECall(UT, MANUAL, -, LOCAL_CNF_REQUIRED);
       
    //@siclog "Step 4A-4R" siclog@
    f_NR5GC_EmergencyCallLimitedService(nr_Cell1);  // @sic R5-227244 sic@
    
    //@siclog "Step 5" siclog@
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]);
    f_NR5GC_EmergencyCallReleaseMT(nr_Cell1);   // @sic R5-227244 sic@
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS2]);
    
    //@siclog "Step 7" siclog@
    f_NR_RRCRelease(nr_Cell1);  // @sic R5-227244 sic@
    
    //@siclog "Step 8" siclog@
    f_Delay(5.0);
    
    //@siclog "Step 9" siclog@
    f_UT_InitiateECall(UT, TEST , -, LOCAL_CNF_REQUIRED);
    
    //@siclog "Step 10" siclog@
    //Check: Does UE send an RRCConnectionRequest on NGC Cell A within 120 seconds?
    f_NR_CheckNoRRCSetupReq(nr_Cell1, v_Wait, -, "Step 10: RRCSetupRequest received");  // @sic R5-227244 sic@
   }


2.9 Change 9
	Function name
	Function fl_TC_11_5_3_TestBody 


	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_InitiateECall is followed by a timer starting

	TTCN module
	ECall_IRAT_NR5GC_NR.ttcn




	  function fl_TC_11_5_3_NR5GC_TestBody() runs on NR_BASE_PTC
  {
    //@siclog "Step 1" siclog@
    //The SS configures:
    // - NR Cell 1 as "Serving cell"
    f_NR_SetCellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);//@sic R5-235303 sic@
    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger("Change Power Level"));
    // @siclog "Step 2" siclog@
    // The UE is Switched ON
    f_UT_SwitchOnUE(UT);

    // @siclog "Step 3" siclog@
    // Wait for 60s for the UE to enter 5GMM-DEREGISTERED.eCALL-INACTIVE state.
    f_Delay(60.0);
    // @siclog "Step 4" siclog@
    //An automatic eCall is initiated. (Note 1)
    f_UT_InitiateECall(UT, AUTOMATIC);
    //@siclog "Step 29" siclog@
    //The UE is switched on.
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Change Power Level");
    f_UT_SwitchOnUE(UT);
    //@siclog "Step 30" siclog@
    //Wait for 60s.
    f_Delay(60.0);
    //@siclog "Step 31" siclog@
    //An automatic eCall is initiated. (Note 1)
    f_UT_InitiateECall(UT, AUTOMATIC , -, LOCAL_CNF_REQUIRED);
    // @siclog "Step 32" siclog@
    // Check: Does the UE transmit an RRCConnectionRequest message in the next 30 sec?// prose CR needed.
    f_NR_CheckNoRRCSetupReq (nr_Cell1, 30.0,-,"step 32 failed");
  }


2.10 Change 10
	Function name
	Function fl_TC_11_3_1_TestBody


	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestIMSCall is followed by a timer starting

	TTCN module
	UAC_NR5GC.ttcn




	  function fl_TC_11_3_1_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var float v_TimerValue := 21.0; //@sic R5-221459 sic@
    timer t_WaitTime;
    
    
    //@sic R5s211658 sic@
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1, //@sic R5-221459 sic@
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(4,1),
                                         -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '0000000'B)));
    f_NR_ModifySysinfo(nr_Cell1);
    f_NR_WaitModificationPeriods(nr_Cell1);

    
    // @sic R5-216192 sic@
    //@siclog "Step 13" siclog@
    //Make the UE attempt a MMTel voice call.
    f_UT_RequestIMSCall(UT , -, LOCAL_CNF_REQUIRED);
     
    //@siclog "Step 14" siclog@
    //Check: Does the UE transmit RRCSetupRequest message including mo-VoiceCall with within 21 s?
    t_WaitTime.start(v_TimerValue);
     alt {
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCSetupRequest(mo_VoiceCall,?)))
        {
          t_WaitTime.stop;
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
        }
      [] t_WaitTime.timeout
        {
        }
    }
   
    //@siclog "Step 15" siclog@
...


2.11 Change 11
	Function name
	Functions fl_TC_11_4_2_TestBody (1 place)
fl_TC_11_4_5_NR5GC_TestBody (1 place)
fl_TC_11_4_6_NR5GC_TestBody (1 place)
fl_TC_11_4_9_NR5GC_TestBody (1 place)

	Reason for change
	A confirmation shall be requested in cases when there the MMI/AT command is followed by the starting of a timer

	Summary of change
	Requested a LOCAL_CNF when the function f_UT_RequestIMSCall is followed by a timer starting

	TTCN module
	Emergency_NR5GC.ttcn




	  function fl_TC_11_4_2_NR5GC_TestBody() runs on NR5GC_PTC
  {

    //@siclog "Step 1" siclog@
    //The SS configures:- "NR Cell 1 as "Serving cell".
    f_NR_SetCellPower(nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    //@siclog "Step 2" siclog@
    //Switch the UE on.
    f_UT_SwitchOnUE(UT);

    //@siclog "Step 3" siclog@
    //Make the UE attempt an IMS emergency call dialling a number which is stored on the ME.
    f_UT_RequestIMSEmergencyCall(UT);
    
    //@siclog "Step 4" siclog@
    //Check: Does the UE performs Generic Test Procedure for IMS Emergency call establishment without IMS emergency registration as specified in TS 38.508-1 [4], subclause 4.9.12?
    f_NR5GC_EmergencyCallLimitedService(nr_Cell1);

    fl_ReleaseCallWithOptionalDeregisterOrRelease(nr_Cell1, 6.0);// @sic R5-223439 R5-235301 sic@
    
    //@siclog "Step 9" siclog@
    //The SS releases RRC connection.
    f_NR_RRCRelease(nr_Cell1);

    //@siclog "Steps 9A" siclog@
    //@sic R5-216198 sic@
    // Make the UE attempt an IMS none-emergency call.
    f_UT_RequestIMSCall(UT , -, LOCAL_CNF_REQUIRED);

    //@siclog "Step 10" siclog@
    //@sic R5-216198 sic@
    //Check: Does the UE transmit an RRCSetupRequest message in the next 30sec?
    f_NR_CheckNoRRCSetupReq (nr_Cell1, 10.0);
   }



